A. Veneziani – Commento alla implementazione di una calcolatrice su 
                          numeri complessi
Premessa

Qualche tempo fa come una delle prime esercitazioni, atte al ripasso di regole e funzionalità relative a Visual Basic è stata proposta l’implementazione di una calcolatrice che permetta le quattro operazioni.

Successivamente si è colta l’occasione dello studio dei files, per realizzare una calcolatrice che effettuasse calcoli che venivano registrati in un apposito file di testo sequenziale (da noi paragonato ad un nastro).
Attualmente invece, si propone lo stesso esercizio della calcolatrice per riflettere in modo più adeguato su alcuni aspetti della programmazione in Visual Basic e specificamente i tipi di dato compositi (ossia i tipi di dato che non sono nativi; si ricorda qui che i tipi nativi sono ad esempio Integer, Boolean ecc.), l’accesso ai singoli campi di un tipo composito, le funzioni, i parametri di tipo non nativo (utilizzati in questa esercitazione per passare dei dati alle funzioni).  

Successivamente, eventualmente, in altra dispensa, si considererà la medesima problematica applicata alle questioni relative alla programmazione ad oggetti, in quanto un numero complesso può essere visto anche come un oggetto e Visual Basic permette di creare semplici oggetti.
Dichiarazione di un nuovo tipo
Come già visto durante l’uso dei files ad accesso casuale, e qui ripreso, nei linguaggi di programmazione (Visual Basic compreso), non è opportuno prevedere l’utilizzo solo e soltanto di variabili di tipo nativo, ossia strutturate come singolo intero (Integer), numero in virgola mobile (Single), booleano (Boolean) ecc, ma talvolta si rende necessario programmare usando strutture dati più complesse che vengono gestite come se fossero un nuovo tipo.  E’ quindi possibile dichiarare tipi compositi che abbiano singoli campi che potranno essere definiti come tipi elementari o tipi a loro volta compositi.  Spesso parlando di tali tipi si parla di strutture.  E ovviamente la definizione di nuovi tipi porta alla conseguenza che possano essere definite variabili strutturate secondo questi nuovi tipi di dati, ossia variabili composite

Ad esempio volendo indicare in una sola tipo le coordinate di un punto grafico sullo schermo (dato composito formato da due interi), si potrà dichiarare la struttura:
Type CoordPunto

      X As Integer

      Y As Integer

End Type

Mentre un nuovo tipo Triangolo, con la stessa logica, potrà essere definito come:
Type Triangolo

     P1 As CoordPunto

     P2 As CoordPunto

     P3 As CoordPunto

End Type

Ovviamente osserviamo che CoordPunto è un tipo non nativo creato e definito precedentemente.
In modo analogo nel nostro esercizio di implementare una calcolatrice che operi su numeri complessi andremo a definire, per comodità, un tipo complesso in questo modo:
Public Type Complesso

    R As Single

    C As Single

End Type
ove R e C sono numeri in virgola mobile rappresentanti la parte reale e immaginaria del numero numero complesso.
Nel  caso del listato presentato questa dichiarazione è indicata in un modulo .BAS apposito, ove sono state inserite anche le funzioni che permettono lo svolgimento delle operazioni sui numeri complessi.  L’indicazione Public che precede la dichiarazione del nuovo tipo Complesso, si riferisce appunto a tale questione, ossia permette che tale tipo non sia utilizzato solo all’interno del modulo .BAS, ma anche all’interno del codice della form, che accoglie l’interfaccia della calcolatrice ed il relativo codice legato agli eventi.   
L’interfaccia

Il progetto definisce ovviamente anche una opportuna interfaccia.  L’interfaccia grafica utilizzata è praticamente analoga a quella della prima versione della calcolatrice, viene qui riproposta.  Oviviamente è possibile alterare in modo più o meno significativo questa interfaccia di partenza, realizzando altre versioni della calcolatrice con differenze solo estetiche oppure anche funzionali.
Abbiamo suddiviso le caselle per inserire i complessi in due parti, così che le caselle stesse non contengano un dato composito successivamente da suddividere (tipo parte intera e immaginaria assieme, ma lascia questo compito all’utente semplificando quindi il compito del programmatore all’origine (si tenga dunque conto che anche le scelte relative alla GUI, a seconda dei casi possono rendere più o meno complicato realizzare un programma).
Ogni pulsante per la singola operazione è collegato ad una apposita routine evento, ovviamente.  Ad esempio per la somma risulta il codice:
Private Sub cmd_Somma_Click()

Dim C1 As Complesso, C2 As Complesso, Ris As Complesso    
    C1.R = CSng(txt_R1.Text)

    C1.C = CSng(txt_C1.Text)

    C2.R = CSng(txt_R2.Text)

    C2.C = CSng(txt_C2.Text)

    Ris = SommaC(C1, C2)

    txt_Rr.Text = Ris.R

    txt_Cr.Text = Ris.C

End Sub
Cosa vuol dire questo codice ?  E perché è scritto così ?
Innanzitutto notiamo che il “contenitore” più esterno è una subroutine evento, legata al cilick sul punsante somma.  Il nome del pulsante è persino deducibile dal codice stesso; esso si chiama cmd_Somma, e il nome della Sub diviene quindi in modo automatico cmd_Somma_Click.   In questa routine sono dichiarate tre variabile di tipo complesso Complesso, che servono ad effettuare l’operazione di somma di numeri complessi voluta. Ossia:

Dim C1 As Complesso, C2 As Complesso, Ris As Complesso    
Questa dichiarazione non sarebbe stata possibile (comunque sarebbe stata priva di senso) se non si fosse definito prima il tipo Complesso, visto in precedenza.  Le variabili C1 e C2 hanno il compito di acquisire il contenuto dei due operandi sulle box esterne.

Le istruzioni per fare ciò sono:

    C1.R = CSng(txt_R1.Text)

    C1.C = CSng(txt_C1.Text)

    C2.R = CSng(txt_R2.Text)

    C2.C = CSng(txt_C2.Text)

In esse di nota la presenza di una funzione di conversione CSng che ha la funzione di trasformare il testo da stringa a numero con la virgola.  Ogni box manda il suo dato (ad esempio parte reale primo numero) nella opportuna parte della variabile composita C1 o C2; l’accesso ai relativi sottocampi (campo reale di tipo Single R), è dato dalla sintassi <variabile composita>.<nome campo>

Successivamente sempre nel medesimo frammento di codice è presente la chiamata alla funzione che effettua l’effettivo calcolo della somma dei due numeri complessi:
    Ris = SommaC(C1, C2)

La funzione, come evidente, ha due parametri e ha nome SommaC.  La sua dichiarazione ed il suo codice è descritto nel modulo .BAS dove è stato dichiarato anche il tipo di dato composito.  Si ricorda che la funzione ha dichiarazione del tipo:
Function SommaC(N1 As Complesso, N2 As Complesso) As Complesso

Da ciò risulta immediatamente che la funzione ha due parametric di tipo composito (tipo Complesso) e che essa rende un valore non di tipo nativo, ma anch’esso composito (tipo Complesso).   Il fatto che la funzione rende un valore di tipo composito, obbliga il programmatore a dichiararla comunque in un modulo .BAS pena il non funzionamento del programma e l’apparire del messaggio d’errore sottostante:
Il codice di questa specifica funzione risulta:
Function SommaC(N1 As Complesso, N2 As Complesso) As Complesso

Dim Ris As Complesso

     Ris.R = N1.R + N2.R

     Ris.C = N1.C + N2.C

     SommaC = Ris

End Function
Come si può osservare i numeri complessi (variabili strutturate) che vengono passate alla funzione, vengono successivamente utilizzate in relazione ai singoli campi R e C, seguendo le ben note regole della somma di variabili complesse, ossia si somma parte intera con parte intera e parte immaginaria con immaginaria.
Questi risultati convergono in Ris, ed è questa variabile che viene associata a quanto la funzione rende (risultato di tipo strutturato o composito che dir si voglia).   Dalla parte di codice che ha chiamato sommaC, il risultato viene raccolto da una apposita variabile di tipo composito, indicata nel listato con Ris.
Ottenuto il risultato non resta altro che renderlo visibile tramite uno spostamento nelle textbox di uscita dei dati e quindi tramite il codice:

    txt_Rr.Text = Ris.R

    txt_Cr.Text = Ris.C

In questo modo l’operazione per effettuare la somma di un numero complesso è davvero conclusa.

Le altre operazioni su complessi sono del tutto analoghe nei meccanismi e nelle premesse concettuali, differendo solo dalla effettiva operazione svolta.
Ad esempio la moltiplicazione tra due numeri complessi equivale allo svolgimento del prodotto:

(a + ib) * (c + id) = a * c + i * a * d + i * b * c + (i)2 b * d = 
(a * c – b * d) + i ( a * d + b * c)
Ricavando, tramite il conto indicato sopra, ed implementato opportunamente in Visual Basic come:

         Ris.R = N1.R * N2.R - N1.C * N2.C

     Ris.C = N1.C * N2.R + N2.C * N1.R

la parte reale e immaginaria.

In modo analogo per tutte e 4 le funzioni previste di somma, sottrazione, moltiplicazione e divisione.

